# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Clément DEMOULINS <clement@archivel.fr>

pkgname=arandr
pkgver=0.1.11
pkgrel=2
pkgdesc="Provide a simple visual front end for XRandR 1.2."
arch=('any')
url="https://christian.amsuess.com/tools/arandr/"
license=('GPL3')
depends=('gtk3' 'python-cairo' 'python-gobject' 'xorg-xrandr')
makedepends=('python-docutils')
source=(https://christian.amsuess.com/tools/$pkgname/files/${pkgname}-$pkgver.tar.gz
        dont-compress-man-pages.patch)
sha256sums=('e4cbbe3698bb812b395770870174be0094bbaeb391105a811f95f42eb182ae02'
            'eb954b763bd2dd7f98ff04a4d45e59db9f8a16cc7384b47ffed733781057ab37')

prepare() {
  cd "$srcdir"/$pkgname-$pkgver
  # Don't create gzip files with a build date in them which makes it unreproducible
  patch -Np1 -i ${srcdir}/dont-compress-man-pages.patch
}

package() {
  cd "$srcdir"/$pkgname-$pkgver
  python setup.py install --prefix=/usr --root="$pkgdir"
}
